
Introduction

Ce document a pour but de reprendre et expliquer brièvement les diverses modifications apportées aux 
programmes initialement conçus pour une calculatrice HP-48G et fournis par le responsable du service travaux 
d’une administration communale. Je présente ensuite la nouvelle structure des répertoires contenant les divers 
programmes. Chaque module est alors illustré par un schéma et un tableau montrant les variables utilisées comme
données (entrées) ou résultats (sorties). Ce n’est donc pas un mode d’emploi des différents programmes, mais 
avec tous les détails fournis, leur fonctionnement en devient implicite. Pour l’utilisation de la calculatrice 
proprement dite, je vous réfère au « Manuel de l’utilisateur ».

1 Principes généraux
En partant des programmes originaux (lorsqu’ils existent, j’en ai écrit de nouveaux), j’ai traduit ou écrit les 
nouveaux programmes en plusieurs étapes :

1. Traduction stricte des programmes tout en opérant une première optimisation, consistant essentiellement 
en :

a. Remplacement des commandes ‘variable’ RCL par variable. Dans les calculatrices 
graphique HP, il suffit de spécifier le nom d’une variable pour en rappeler le contenu sur la pile, s’il 
s’agit d’une donnée, ou pour exécuter le programme si la variable contient un programme.

b. Création de deux premières fonctions très fréquemment utilisées 
i. T->G : traduit les angles de trigonométriques (zéro à droite ou à l’est), en « géométrique » 

ou gisement (zéro au nord)
ii. TGG : « tangente géométrique », qui applique une petite correction (avant ou après le 

calcul de la tangente) pour considérer les angles donnant une valeur infinie (si j’ai bien 
compris).

c. Remplacement du mode d’introduction des données, initialement avec des libellés (TAG) sur la pile
et utilisation de la fonction INPUT, par un formulaire d’entrée de données activé par la fonction 
INFORM. Dans celui-ci, les valeurs par défaut sont celles existant précédemment : contrairement 
aux anciens programmes, les variables n’ont pas une valeur initiale nulle. La fonction INFORM 
prévoit une option de réinitialisation des variables (champs), que j’ai prévue à 0.

d. Suppression de la suppression des variables intermédiaires (PURGE), et du tri des variables ou 
programmes restants (ORDER) à la fin de la plupart des programmes. Cette dernière opération 
étant particulièrement coûteuse en temps de calcul. 

e. La nouvelle version de ZERO a pour but d’initialiser les variables (avec comme valeur initiale 0) et 
de trier les programmes et variables. Cette opération ne devant s’effectuer qu’exceptionnellement.

2. Dans une seconde étape, j’ai optimisé les programmes pour :

a. Utiliser un stricte minimum de variables intermédiaires : réorganisation des calculs, jeu sur la pile, 
utilisation de variables locales (sorte de programme imbriqué, dont la structure est : 
 -> a b c << code avec les variables locales a b et c >> )
La plupart des résultats intermédiaires ne sont donc plus disponibles, j’ai conservé essentiellement
ceux qui étaient «étiquetés» sur la pile.

b. Utilisation d’un plus grand nombre de sous-programmes ou fonctions revenant de manière très 
récurrentes (voir descriptif ci-après).

3. Finalement, j’ai ajouté des tests et messages d’erreur pour les situations anormales (points identiques, 
droites parallèles, cercles disjoints, corde supérieure au diamètre, point à l’intérieur du cercle, …)

4. Modification du nom de certains points et variables, afin de lever certaines ambiguïtés

2 (Ré-) Installation et configuration

2.1 Installation
Cette section présente les diverses manières d’installer les programmes dans la calculatrice. Pour les détails, 
référez-vous à l’ Annexe A : Procédures d’installation 

2.1.1 À la main
La première option est d’encoder tous les programmes à la main, directement sur la calculatrice, mais cette 
opération est fastidieuse et comporte un grand risque d’erreurs d’encodage. A n’utiliser qu’en dernier ressort !

2.1.2 À partir d’un PC
Pour installer les programmes à partir d’un PC, vous devez disposer d’une connexion USB, et du programme 
Conn4x (disponible sur le site HP). Il faut créer la structure des répertoires et sous répertoires à la main, et ensuite 
y transférer les programmes. 

HOME
+ GEOM
  + Droites
  + Cercles
  + Surface
  + XYZ

Voir la section 4 "Code des programmes et variables CST" pour plus de détails.

2.1.3 À partir de la carte SD
Il est également possible de copier les programmes depuis le PC vers une carte SD, et d’ensuite insérer cette carte
dans la calculatrice, et d’y effectuer le transfert à la main (via le menu FILES). Référez-vous au manuel de la 
calculatrice pour plus de détails (je n’ai personnellement pas investigué / testé cette option).

Une fois les programmes installés dans la mémoire principale (mémoire RAM, ou « vive ») de la calculatrice, il est 
possible de créer une copie de sauvegarde de l’entièreté de la mémoire via la fonction ARCHIVE, et d’ensuite la 
restaurer au moyen de RESTORE. Cette opération « écrase » toutes les données de la calculatrice, et pas 
uniquement celles relatives à ces programmes. C’est pourquoi je n’ai pas suivi cette option, mais si vous voulez 
suivre cette voie, référez-vous au manuel.

Il est également possible de sauvegarder uniquement un répertoire (et ses sous-répertoires), un programme ou 
une variable isolée sur la mémoire FLASH (mémoire permanente) ou sur une carte SD. C’est la méthode que j’ai 
choisie, et ai donc créé une copie de sauvegarde « prête à l’emploi » dans ces deux mémoires (FLASH & SD).

2.2 Configuration de la calculatrice HP 50g
Par rapport à la configuration standard (d’usine), la calculatrice est configurée comme suit :

 Mode RPN : tous les  programmes sont écrits en supposant le mode RPN ( -95 CF).
 Séparateur de décimales : ‘,’ ( -51 SF).
 Mode Gradian.
 Mode approximatif, requis pour certaines opérations, correspond au mode de la HP-48G ( -105 SF).
 Coordonnées rectangulaires (-15 CF -16 CF) : mode modifié dans les programmes pour les 

conversions de coordonnées rectangulaires en coordonnées polaires ou sphériques, et vice-versa.
 Mode utilisateur (User) activé (-61 SF -62 SF)  : afin d’avoir le menu (CUSTOM) sur la touche [APPS]. 
 Menus ‘Soft’ ( -117 SF) : facultatif, par facilité / préférence pour naviguer dans les menus.

Cette configuration est activée par le petit programme CFG, disponible dans le répertoire GEOM, qui (ré-)initialise 
également toutes les variables et crée le menu principal (dans HOME).

3 Modules et interfaces

3.1 Menus 
Puisque dans cette version je ne supprime pas les variables, afin de garder une facilité d’accès aux programmes et
pour pouvoir naviguer plus facilement entre les modules (chaque groupe de programmes étant dans un répertoire 
distinct), j’ai utilisé la fonction CUSTOM dont le but est de fournir un menu utilisateur, associant une action aux 
touches de la première rangée : touches fonctions de la calculatrice. 

La fonction CUSTOM repose sur la variable CST (cherchée dans le répertoire courant, ou dans un supérieur), que 
j’ai donc créée dans chaque répertoire. Pour modifier le menu, l’ordre des fonctions ou en
ajouter / supprimer, il suffit d’éditer la liste contenue dans la variable CST.

Les menus débutent par les fonctions de navigation (remontée au répertoire GEOM,
descente dans un sous-répertoire, …), suivies des programmes du module courant (que
ces programmes soient spécifiques et dans le sous-répertoire ou communs et dans le
répertoire principal), viennent ensuite certaines fonctions de conversions (du répertoire principal). Les menus se 
terminent par les fonctions d’initialisation (ZERO, CFG), accessibles plus rapidement par [APPS] (en mode USR 
pour activer le menu) et [PREV] ([Maj. Gauche] [NXT]).

3.2 Saisie des données
Précédemment, la saisie des données se faisait via une « invite avec chaîne d’entrée » : chaine de caractères et 
liste de variables « étiquetées », arguments de la fonction INPUT.  Dans la nouvelle version, la saisie se fait par 
des « formulaires d’entrée », activés par la fonction INFORM, qui présente un titre et une
liste de champs, chacun avec une valeur par défaut (que j’ai mise à la valeur actuelle de la
variable correspondante), une valeur de réinitialisation (spécifiée à zéro pour tous) et un
petit texte d’aide (affiché au bas de l’écran). Le passage d’un champ à l’autre se fait par les
flèches directionnelles. 

La (nouvelle) valeur d’un champ peut simplement être entrée, mais on peut également l’éditer ([EDIT]). Le menu 
des formulaires comprend les fonctions d’annulation ([CANCL] (ou la touche [CANCEL] en bas du clavier)), de 
validation ([OK] (ou [ENTER])), et en appuyant sur [NXT],  on accède au second menu qui permet la 
réinitialisation ([RESET]), de voir le type de donnée à fournir ([TYPES]), et ce revenir en mode calcul ([CALC]).  
Le mode calcul met la valeur courante du champ au niveau 1 de la pile, qui peut donc servir de base pour calculer 
une nouvelle valeur. En mode calcul, l’entête (premières lignes de l’écran) reprend le titre du formulaire et le texte 
d’aide du champ en question. Une pression de [STS] permet de visualiser les lignes d’état de la calculatrice. 
Lorsqu’on active un autre menu (par exemple le menu utilisateur), l’entête rappelle qu’il faut presser sur [CONT] 

([Maj. Gauche] [ON]) pour revenir au menu spécifique du mode calculatrice.  Ceci est indispensable, car la 
sortie de ce mode se fait par annulation ([CANCL] (ou [CANCEL])) ou plus probablement par la validation de 
l’entrée via [OK] (et non [ENTER] !) qui prend la valeur au niveau 1 de la pile.  Cette faculté peut être utilisée pour
par exemple entrer l’angle gamma mesuré (voir plus loin), et le convertir en gisement par la fonction ->G.  Ainsi, à 
partir du programme et formulaire AGt du module Droites,  pour introduire le gisement gA à partir de l’angle 
mesuré , il "suffit" de :
- presser sur [NXT] [CALC] pour revenir en mode calcul sur la pile, y entrer l’angle mesuré au rang 1 de la pile,
- presser sur [APPS] [NXT] [->G] pour convertir l’angle mesuré en gisement, et finalement,
- presser sur [CONT] [OK] pour valider l’entrée et revenir dans le formulaire.

3.3 Structure des répertoires 
Les programmes communs et fonctions se trouvent dans le répertoire principal GEOM , les modules étant dans les 
divers sous-répertoires :

 Structure arborescente des répertoires et fichiers : nom sur HP50g [nom sur PC]

HOME
+ GEOM
  | AB
  | AGt
  | AP
  | CFG
  | CST
  | CSTHOME
  | G→R [G^141R]
  | in&sto
  | init
  | INST
  | msgerr
  | P→R [P^141R] 
  | REST
  | R→G [R^141G] 
  | R→P [R^141P] 
  | R→S [R^141S] 
  | sto&tag
  | S→R [S^141R] 
  | TGG
  | T→G [T^141G] 
  | γ→G [^145^141G]
 (gammaG)
  + Cercles
    | CST
    | EchC
    | IJK
    | IJR
    | Int2C
    | IntDC

    | SECT
    | TgC
    | ZERO
    | →OTJ [^141OTJ]
(GEOM continued)
  | …
  + Droites
    | CD
    | CGt
    | cM
    | CST
    | DIST
    | LEVE
    | POINT
    | ZERO
  + Surface
    | CST
    | SURF
    | SURFM
    | ZERO
  + XYZ
    | BASE
    | CST
    | gCIBLE
    | XYCIBLE
    | XYZCIBLE
    | ZERO
    | γCIBLE [^145CIBLE]
 (gammaCIBLE)

 Répertoire principal GEOM :
o Les deux programmes communs à plusieurs modules (AB & AGt) et les programmes de 

configuration (AP, ZERO, CFG, REST & INST) : 
 AB : saisie des points A et B, et calcul du gisement de la droite AB.
 AGt : saisie du point A et du gisement de la droite passant par A.
 AP : saisie des coordonnées de l’instrument au point A,  ainsi que des coordonnées du 

point de référence P : l’angle mesuré par rapport au zéro de l’instrument (AP (gammaAP)) 
et le gisement gAP. Si les coordonnées XY du point P sont fournies, c’est-à-dire si elles 

P© Pierre Hardy 2010 - 2021 Page 6 / 21 26/06/21



Programmes de Géomètre pour Calculatrice HP 50g Version 1.3b

sont différentes de (0 , 0), celles-ci seront utilisées pour calculer (et remplacer) le gisement
gAP.

 ZERO : initialisation (et création si besoin) des variables à 0, et tri.
 CFG : configuration initiale de la calculatrice, selon le point 2.2 ci-dessus, et exécute les 

programmes ZERO du répertoire GEOM et de ses sous-répertoires.
 REST : restaure le répertoire à partir d’une sauvegarde enregistrée (par INST) dans la 

mémoire FLASH (permanente) de la HP 50g, et reconfigure la calculatrice via CFG.
Ce programme efface le répertoire GEOM avant de le re-créer et d’y copier les 
programmes, toutes les données s’y trouvant seront donc perdues. Cette procédure 
est également directement disponible dans la mémoire FLASH, et peut être exécutée à 
partir de cet endroit (voir section 2 : (Ré-) Installation)

 INST : (ré)installe les programmes GEOM tant dans la mémoire principle que dans la 
mémoire FLASH à partir de la carte mémoire SD. 
Ce programme efface le répertoire GEOM  et la sauvegarde dans la mémoire FLASH, 
avant de les re-créer et d’y copier les programmes, toutes les données s’y trouvant 
seront donc perdues. Cette procédure est également directement disponible sur la carte 
SD, et peut être exécutée à partir de cet endroit (voir section 2 : (Ré-) Installation)

o Les fonctions, disponibles dans la plupart des menus, sont utilisables directement et notamment 
lorsqu’on revient en mode calcul depuis un formulaire de saisie ([CALC]). Elles sont également  
fréquemment utilisées dans les programmes.  :

 ->G (gamma->G) : convertit l’angle  mesuré au niveau 1 de la pile en gisement (entre 
0 et 400 grades), selon les données gAP et AP (gammaAP) du point de référence P, 
entrées via AP. 

 R->P : convertit les deux nombres au niveau 1 & 2 de la pile des coordonnées 
rectangulaires en coordonnées polaires.

 P->R : convertit les deux nombres au niveau 1 & 2 de la pile des coordonnées 
polaires en coordonnées rectangulaires.

 R->G : convertit les deux nombres au niveau 1 & 2 de la pile des coordonnées 
rectangulaires en coordonnées  géométrique (polaires où l’angle est converti en 
gisement).

 G->R : convertit les deux nombres au niveau 1 & 2 de la pile des coordonnées 
géométrique en coordonnées  rectangulaires.

 R->S : convertit les trois nombres au niveau 1, 2 & 3 de la pile des coordonnées 
rectangulaires (ou cartésiennes) en coordonnées sphériques.

 S->R : convertit les trois nombres au niveau 1, 2 & 3 de la pile des coordonnées 
sphériques en coordonnées rectangulaires.

 T->G : convertit l’angle trigonométrique en gisement.
 TGG : « tangente géométrique » : applique une petite correction (avant ou après le calcul 

de la tangente) pour considérer les angles donnant une valeur infinie (si j’ai bien compris). 
Notez que la HP 50g peut automatiquement « convertir » les résultats infinis en +/- 
9.999999E499 (voir manuel et indicateurs systèmes -21 et -22). Ne disposant pas de 
données de test, je n’ai pas exploré cette voie.

o Les sous-programmes créés (par convention personnelle, tous les sous-programmes sont dans 
des variables dont le nom commence par une lettre minuscule) :

 init : initialise la première liste de variables (et les crées automatiquement si elles 
n’existent pas encore), trie les variables selon les deux listes concaténées et efface la pile.
Ce sous-programme est appelé par les programmes ZERO.

 in&sto: entrée et mise en mémoire des données (INFORM et STO), utilisé par tous les 
programmes nécessitant une saisie de données. Ce programme efface également la pile 
et reconfigure la calculatrice en mode Gradian et Cartésien.  Si vous ne souhaitez pas que
la pile soit effacée à chaque saisie, vous pouvez supprimer la fonction ‘CLEAR’ de celui-ci.

 sto&tag : met la valeur au niveau 1 de la pile en mémoire et garde une copie étiquetée 
(TAG) sur la pile : l’étiquette étant la même que le nom de la variable

 msgerr : affiche un message d’erreur (MSGBOX), efface la pile et termine le programme 
en cours (KILL)

o CST : variable (liste) contenant les éléments du menu, activé par la fonction CUSTOM, également 
assignée sur la touche [APPS] en mode USER. Si vous souhaitez changer l’ordre des 
programmes, ajouter ou supprimer des éléments (fonctions de conversions par exemple, … ), vous
pouvez modifier cette liste.  Dans les sous-répertoires, ce menu reprend également des 
programmes et fonctions communes, tels que AB, AGt, R->P, … , présents dans le répertoire 
principal GEOM, ce qui permet un accès aisé à ceux-ci, comme s’ils étaient dans les sous-
répertoires.

o Les sous-répertoires suivants, par groupe de programmes :
 Droites
 Cercles
 Surface
 XYZ

Ces répertoire (GEOM) et sous-répertoires doivent être créés manuellement.

 Module Droites :
o CD : saisie des points C et D, et calcul du gisement de la droite CD, et du point d’intersection M.
o CGt : saisie du point C et du gisement de la droite passant par C, et calcul du point M.
o cM : sous-programme, calcule le point d’intersection M (appelé par CD et CGt).
o POINT : saisie de la distance ‘a’ sur la droite AB, et calcul des coordonnées du point F sur la 

droite AB.
o DIST : saisie du point E, et calcul des distances ‘a’ (sur la droite AB) et ‘b’ (perpendiculairement), 

ainsi que des coordonnées du point ‘F’ sur la droite AB.
o LEVE : saisie des distances ‘a’ (sur la droite AB) et ‘b’ (perpendiculairement), et calcule les 

coordonnées des points E (« levé »)  et F (sur la droite AB).
o ZERO : initialisation (et création si besoin) des variables à 0, et tri.
o CST : variable (liste) contenant les éléments du menu.

 Module Cercles :
o IJK : saisie des points I, J et K, et calcule le centre et le rayon du cercle à partir de ces 3 points.
o IJR : saisie des points I, J et du rayon R, et calcule le centre du cercle à partir de ces 2 points et 

du rayon.
o Int2C : saisie des centres O et O2, et des rayons R et R2, et calcule les 2 points d’intersection 

des 2 cercles.
o TgC : saisie du point T sommet des tangentes, des coordonnées du cercle (O et R), et calcule les 

tangentes au cercle à partir du point T.
o IntDC : saisie du centre O et du rayon R du cercle, et calcule l’intersection d’une droite (dont les 

coordonnées ont été préalablement entrées via AGt ou AB) et du cercle.
o SECT: saisie du rayon R et de la corde c, et calcule les éléments du cercle

coupé par une corde. Les résultats sont étiquetés d’un message plus
explicite, avec le nom de la variable correspondante entre parenthèses.

o ->OTJ : saisie du point I (première extrémité de la corde) et du point A
(aligné, donnant le gisement de la corde) et sur base des résultats de 
SECT, calcule le centre O du cercle, le sommet T des tangentes, et les coordonnées de l’autre 
extrémité J de la corde.

o EchC : échange les coordonnées du cercle primaire (O et R) avec le cercle secondaire (O2 et 
R2), ce qui permet d’utiliser ensuite ce second cercle dans comme donnée dans les autres 
programmes

o ZERO : initialisation (et création si besoin) des variables à 0, et tri.

o CST : variable (liste) contenant les éléments du menu.

 Module Surface :
o SURF : Calcul de la surface par introduction des points du polygone un à un, tous ces points ainsi 

que les distances entre eux, sont repris dans une variable de type
« matrice » (Polyg). Dans cette matrice, la première colonne correspond
aux abscisses, la deuxième aux ordonnées des points et la troisième
reprend la distance calculée par rapport au point précédent. 

o SURFM : Calcul de la surface en prenant les coordonnées dans la matrice 
Polyg. Si le niveau 1 de la pile contient une matrice, elle sera utilisée
comme donnée et enregistrée dans Polyg .  Et si le dernier point n’est pas égal au premier, une 
ligne est ajoutée pour « fermer » le polygone. Ceci permet de corriger les coordonnées entrées 
avec SURF ou de créer directement une matrice (n x 2 (ou n x 3)) de points par l’éditeur de matrice 
de la HP 50g (MTRW). Si la matrice Polyg initiale ne contient que 2
colonnes, une troisième sera ajoutée pour enregistrer les distances.

o ZERO : initialisation (et création si besoin) des variables à 0, et tri. 
o CST : variable (liste) contenant les éléments du menu.

 Module Coordonnées XYZ :
o BASE : saisie des coordonnées de l’instrument au point A,  ainsi que des coordonnées du point de 

référence P : l’angle mesuré par rapport au zéro de l’instrument (gammaAP AP) et le gisement 
gAP. Si les coordonnées XY du point P sont fournies, c’est-à-dire si elles sont différentes de (0 , 0),
celles-ci seront utilisées pour calculer (et remplacer) le gisement gAP. Notez que les coordonnées 
XY du point A, ainsi que les coordonnées du point P (Xa, Ya, Xp, Yp, gAP et AP) sont stockées 
dans le répertoire GEOM, et non dans le sous-répertoire XYZ, de sorte que BASE et AP (dans GEOM) 
partagent les mêmes variables, également utilisées par la fonction commune  ->G.

o gCIBLE : saisie du gisement gA de la droite AB, de l’angle zénithal (thêta Θ), de la distance 
oblique (d), de l’hauteur de la canne (Hb) au point B (cible), et calcul des coordonnées X, Y et Z du
point B.

o CIBLE (gammaCIBLE): saisie de l’angle de la droite AB mesuré par rapport au zéro de 
l’instrument (gammaA A), de l’angle zénithal (thêta Θ), de la distance oblique (d), de la hauteur 
de la canne (Hb) au point B (cible), et calcul des coordonnées X, Y et Z du point B.

o XYCIBLE : saisie de l’abscisse Xb et de l’ordonnée Yb du point B, de l’angle zénithal (thêta Θ), 
de la distance oblique (d), de la hauteur de la canne (Hb) au point B (cible), et calcul du gisement 
gA de la droite AB sur base des coordonnées XY fournies, et calcul des coordonnées X, Y et Z du 
point B sur base du gisement, de l’angle zénithal et de la distance oblique. Les coordonnées XY 
calculées (et remises en mémoire) seront différentes de celles introduites si la distance oblique ne 
correspond pas.

o XYZCIBLE : saisie des coordonnées X, Y et Z du point B, de la hauteur de la canne au point B 
(cible), et calcul de la distance oblique (d), du gisement gA de la droite AB et de l’angle zénithal 
(thêta Θ).

o ZERO : initialisation (et création si besoin) des variables à 0, et tri. 
o CST : variable (liste) contenant les éléments du menu.

3.4 Schémas, entrées et sorties

Schémas

Pour chaque jeu de programmes réunis dans un sous-répertoire, voici un schéma illustrant toutes les variables 
utilisées (en entrée ou en sortie).  Les exemples étant cotés, ceux-ci peuvent également servir de « jeu de test » 
pour les divers programmes.

Entrées et sorties

Les variables n’étant pas initialisées (à zéro) au début de chaque programme, et les
« formulaires de saisie » prenant les valeurs actuelles comme valeur initiales, on peut
également utiliser ces programmes pour visualiser les données, éventuellement produites
par d’autres programmes. Il suffit d’appuyer sur [CANCEL] pour sortir du formulaire sans
aucune opération.  Ainsi dans le répertoire Droites, on peut revoir le résultat de LEVE
avec DIST, et vice-versa, l’un prenant en effet les résultats de l’autre comme entrée.

Sous le schéma, un tableau reprend la liste des programmes et la liste des variables
utilisées.  A l’intérieur du tableau, les codes suivants déterminent s’il s’agit d’une donnée,
d’un résultat, … :

- E : entrée, à fournir via un écran de saisie (formulaire).
- S : sortie avec « étiquette » (TAG) disponible sur la pile, l’étiquette correspondant au nom de la variable où 

est enregistré ce résultat.
- R : autre résultat (non publié sur la pile), uniquement stocké en mémoire.
- P : variable pré-requise (saisie ou calculée par un autre programme accessible via le menu : dans le même

répertoire ou dans le répertoire supérieur (GEOM) où résident les fonctions communes). 

3.4.1 Programmes communs

Pgrm/
Fct Xa Ya Xp Yp gAP AP Xb Yb gA dAB
AP E E E E E/S E
G P P
AB E E E E S S
AGt E E E

3.4.2 Module ‘Droites’

Pgrm X
a

Y
a

X
b

Y
b

g
A

dA
B

X
c

Y
c

X
d

Y
d

g
C

dC
D

X
m

Y
m

X
e

Y
e

X
f

Y
f

a b

AB E E E E S S
AGt E E E
CD P P P E E E E S S S S
CGt P P P E E E S S
POIN
T P P P S S E
DIST P P P E E S S S S
LEVE P P P S S S S E E

x

y

(0 , 0)

(10 , 0)

(0 , 8)

F (5 , 3)

A (-4 , -3)

M (2 , 1)

B (8 , 5)

gA = 
62.57

E (3 , 6)

C (-2 , 3)

D (10 , -3)

gC = 
129.52 dAB = 14.42

a = 10.82

b = 3.61

dCD = 13.42

3.4.3 Module ‘Cercles’

x

y

(0 , 0)

R 4.03
R2 4.03

(10 , 0)

(0 , 8)

J (5 , 5)

A (-1 , -4)

I (1 , -1)

B (7 , 8)

t =8.06
T (-3 , 6)

O (4.5 , 1)

c 
= 

7.
21

b = 4.98

O2 (1.5 , 3)

K (8.5 , 0.5)

f = 2.23

l = 9.01

ST = 
17.99SG = 

11.49

gOJ = 
7.92

α = 
140.97

a = 8.93

gOT = 
337.43

gOI = 
266.95

gTO = 
137.43

gTJ = 
107.92

gTI = 
166.95

gA = 
37.43

dA
B 

= 
14

.4
2

Pgrm Xi Yi Xj Yj Xk Yk Xo Yo R c Xa Ya gA Xb Yb dA
B

Xt Yt gO
T

gO
I

gO
J

gT
O

gT
I

gT
J

l t  a b f ST SG Xo
2

Yo
2

R2

IJK E E E E E E S S S R R

IJR E E E E S S E R R S S R

Int2C S S S S E E E S E E E

TgC S S S S E E E R E E S S S S S S S S

AB E E S E E S

AGt E E E

IntDC S S S S E E E S P P P

SECT E E S S S S S S S S

OTJ E E S S S S P P E E R S S P P

EchC E/R E/R E/R E/R E/R E/R

3.4.4 Module ‘Surface’

Pgrm X
a

Ya X
b

Y
b

Poly
g

S

SURF E E E E S S
SURF

M R R R R E/S S

x

y

A (2 , -3) 
= B(7)

B(1) (6 , -1)

B(3) (8 , 3)

B(4) (4 , 5)

B(5) (0 , 3)

B(6) (-1 , -1)

S = 40

B(2) (5 , 2)

3.4.5 Module ‘Coordonnées XYZ’

x (E)

z

(0 , 0)

gA = 
44.23

y (N)

A (5 , 2 , 2)

Ha = 1.5

Hb = 2

B (10 , 8 , 5)

Q = 
73.18

d =
 8.56

Cible

Instrument

w (Zéro instrument)

P (15 , 5 , 3.5)

Hb = 2

Cible

Q = 
87.95

gAP = 
81.45

d = 10.63

AP = 
344.01

A = 
306.80

Nord

Point de 
référence

Pgrm Xa Ya Za Ha Xp Yp
A
P

gA
P A gA Θ d Hb Xb Yb Zb

BASE E E E E E E E E/S
CIBLE P P P P P P E S E E E S S S
gCIBLE P P P P E E E E S S S
XYCIBLE P P P P S E E E E/S E/S S
XYZCIBL
E P P P P S S S E E E E

Notez que les coordonnées XY du point A, ainsi que les coordonnées du point P 
(Xa, Ya, Xp, Yp, gAP et AP) sont stockées dans le répertoire GEOM, et non dans le 
sous-répertoire XYZ, de sorte que BASE et AP (dans GEOM) partagent les mêmes 
variables, également utilisées par la fonction commune  ->G.

4 Code des programmes et variables CST
Les programmes et variables CST (présentes dans chaque sous-répertoire) sont repris avec l’entête HP  (%
%HP: T(3)A(G)F(,);) qui indique au programme de communication ‘Conn4x’ comment transférer et traduire les
caractères spéciaux.  Il suffit donc de copier l’entièreté de chaque programme ou variable, tel quel, dans un fichier 
texte (.TXT) et de les transférer ensuite via Conn4x pour les obtenir sur la HP 50g (une autre méthode consiste à 
utiliser une carte SD).  Vous devrez néanmoins construire la structure des répertoires à la main.

Bon nombre de noms fonctions contiennent une flèche vers la droite (‘->’ : un seul caractère).  Sur un PC, dans le 
nom du fichier, il faut remplacer la flèche par la séquence ‘^141’, ainsi la fonction ‘R->P’ sera dans le fichier 
‘R^141P.TXT’. De même le caractère  (gamma) doit être remplacé par ‘^145’ dans le nom de fichier.

4.1 Programmes communs  (répertoire GEOM)
CSTHOME (variable CST dans le répertoire HOME)
%%HP: T(3)A(G)F(,);
{ GEOM { Droites \<< GEOM Droites \>> }
{ Cercles \<< GEOM Cercles \>> }
{ Surface \<< GEOM Surface \>> }
{ XYZ\<< GEOM XYZ\>> } }

CST
%%HP: T(3)A(G)F(,);
{ UPDIR Droites Cercles Surface XYZ AP \Gg\->G 
R\->P P\->R R\->G G\->R T\->G R\->S S\->R TGG 
AB AGt ZERO CFG REST INST }

AB
%%HP: T(3)A(G)F(,);
\<< "Coordonn\233es droite A - B" 
{ { "Xa" "Abscisse point A" 0 }
{ "Ya" "Ordonn\233e point A" 0 }
{ "Xb" "Abscisse point B " 0 }
{ "Yb" "Ordonn\233e point B" 0 } }
{ Xa Ya Xb Yb } in&sto Xb Xa - Yb Ya - R\->G 
'gA' sto&tag SWAP IF DUP 0 == 
THEN "Points A & B identiques !" msgerr END
'dAB' sto&tag 
\>>

AGt
%%HP: T(3)A(G)F(,);
\<< "Coordonn\233es point A & Gt" 
{ { "Xa" "Abscisse point A" 0 }
{ "Ya" "Ordonn\233e point A" 0 }
{ "gA" "Gisement point A" 0 } { } }
{ Xa Ya gA } in&sto gA 'gA' \->TAG 
\>>

AP
%%HP: T(3)A(G)F(,);
\<< "Coordonn\233es Instruments" 
{ { "Xa" "Abscisse point A" 0 } 
{ "Ya" "Ordonn\233e point A" 0 } 
{ "Xp" "Abscisse point P" 0 } 
{ "Yp" "Ordonn\233e point P" 0 } 
{ "gAP" "Gisement au point A \-> P" 0 } 
{ "\GgAP" "Angle \Gg mesur\233 au point A \-> P" 
0 } } { Xa Ya Xp Yp gAP \GgAP } in&sto
IF Xp Yp + 0 == THEN gAP 
ELSE Xp Xa - Yp Ya - R\->G SWAP DROP END 
'gAP' sto&tag
\>>

->G  (nom du fichier correspondant : ^145^141G.txt)
%%HP: T(3)A(G)F(,);

\<< gAP + \GgAP - IF DUP 0 < THEN 400 + 
ELSE IF DUP 400 > THEN 400 - END END 
\>>

R->P (nom du fichier correspondant : R^141P.txt)
%%HP: T(3)A(G)F(,);
\<< -16 CF \->V2 -16 SF V\-> \>>

P->R (nom du fichier correspondant : P^141R.txt)
%%HP: T(3)A(G)F(,);
\<< -16 SF \->V2 -16 CF V\-> \>>

R->S (nom du fichier correspondant : R^141S.txt)
%%HP: T(3)A(G)F(,);
\<< -16 CF \->V3 -15 SF -16 SF V\-> \>>

S->R (nom du fichier correspondant : S^141R.txt)
%%HP: T(3)A(G)F(,);
\<< -15 SF -16 SF \->V3 -16 CF V\-> \>>

T->G (nom du fichier correspondant : T^141G.txt)
%%HP: T(3)A(G)F(,);
\<< IF DUP 100 \<= THEN 100 ELSE 500 END SWAP - \>>

R->G (nom du fichier correspondant : R^141G.txt)
%%HP: T(3)A(G)F(,);
\<< -16 CF \->V2 -16 SF V\-> IF DUP 100 \<= 
THEN 100 ELSE 500 END SWAP – 
\>>

G->R (nom du fichier correspondant : G^141R.txt)
%%HP: T(3)A(G)F(,);
\<< IF DUP 100 \<= THEN 100 ELSE 500 END SWAP – 
-16 SF \->V2 -16 CF V\-> 
\>>

TGG
%%HP: T(3)A(G)F(,);
\<< 1,E-99 0,000000001 \-> g o p 
  \<<
  CASE g 0 < THEN g 200 + TAN END 
       g 0 == g 200 == OR g 400 == OR 
         THEN g TAN o + END
       g 100 == g 300 == OR THEN g p + TAN END
       g 400 < THEN g TAN END
       g 400 - IF DUP 0 == OVER 200 == OR
         THEN TAN o + ELSE DROP g p + TAN END
       END
  \>> 
\>>

5 Code des programmes originaux (HP-48G)
Pour mémoire, voici le code source fourni, sauvegarde des programmes tournant sur la HP-48G. Certains sont 
incomplets, d’autres simplement inexistants et finalement j’ai décelé l’une ou l’autre petite erreur dans ces sources 
(très certainement corrigées dans la calculatrice elle-même). L’intérêt de ces sources est qu’elles permettent 
percevoir le chemin parcouru, mais aussi et surtout de retrouver la genèse des formules. Ces programmes initiaux 
utilisent beaucoup de variables intermédiaires qui ont le mérite de rendre le code plus lisible. Je liste donc le tout 
tel quel (avec quelques petites corrections tout de même), pour référence. Notez que j’ai supprimé l’entête ‘‘%%HP: 
T(3)A(G)F(,);’, celle-ci étant identique pour chaque programme.

5.1 Intersection
ZERO
\<< 
{ Xa Ya Xb Yb gA dAB \
GDa \GDc \GDx \GDy p o \Ga
gC dCD Xc Yc Xd Yd Ym Xm }
PURGE { AB AGT CD CGT ZERO
} ORDER CLEAR
\>>

AB
\<< 
CLEAR
"Introduire les coord
 de la droite A - B"
{ ":B:()
:A:()" } INPUT GRAD 
-15 CF -16 CF OBJ\-> 
C\->R 'Ya' STO 'Xa' STO
C\->R 'Yb' STO 'Xb' STO 
'Xb' RCL 'Xa' RCL – 
'Yb' RCL 'Ya' RCL – 
\->V2 -16 SF V\->
'\Ga' STO 'dAB' STO
'\Ga' RCL DUP DUP DUP
  CASE 0 \<=
    THEN 3 DROPN
'\Ga' RCL NEG 100 +
    END 100 <
    THEN DROP2 
'\Ga' RCL NEG 100 +
    END 100 ==
    THEN DROP 
'\Ga' RCL NEG 100 +
    END 100 >
    THEN 
'\Ga' RCL 100 - NEG 400 +
    END
  END 'gA' STO 
'gA' RCL 'gA' \->TAG 
'dAB' RCL 'dAB' \->TAG 
{ AB AGT CD CGT ZERO  } 
ORDER
\>>

AGT
\<<
"Introduire les coord
 de A & Gt"
{ ":gA:
:A:()" } INPUT 
GRAD -15 CF -16 CF 
OBJ\-> C\->R 
'Ya' STO 'Xa' STO 'gA' STO
{ AB AGT CD CGT ZERO } 
ORDER 
'gA' RCL 'gA' \->TAG
\>>

CD
\<<
"Introduire les coord
 de la droite C-D"
{ ":D:()
:C:()" } INPUT GRAD
 -15 CF -16 CF OBJ\-> 
C\->R 'Yc' STO 'Xc' STO
C\->R 'Yd' STO 'Xd' STO 
'Xd' RCL 'Xc' RCL – 
'\GDx' STO 'Yd' RCL 
'Yc' RCL - '\GDy' STO '\
GDx' RCL '\GDy' RCL 
\->V2 -16 SF V\->
'\Ga' STO 'dCD' STO
'\Ga' RCL DUP DUP DUP
  CASE 0 \<=
    THEN 3 DROPN
'\Ga' RCL NEG 100 +
    END 100 <
    THEN DROP2 
'\Ga' RCL NEG 100 +
    END 100 ==
    THEN DROP 
'\Ga' RCL NEG 100 +
    END 100 >
    THEN 
'\Ga' RCL 100 - NEG 400 +
    END
  END 'gC' STO 
'gC' RCL 'gC' \->TAG 
'dCD' RCL 'dCD' \->TAG 
{ \GDx \GDy \Ga } PURGE 
'Yc' RCL 'Ya' RCL – 
'\GDy' STO 
'Xc' RCL 'Xa' RCL – 
'\GDx' STO 
'1*1,E-99' \->NUM 
'o' STO 

'1*,000000001' \->NUM 
'p' STO 
'gA' RCL DUP DUP DUP DUP
DUP DUP DUP
  CASE 0 <
    THEN 7 DROPN
'gA' RCL 200 + TAN
    END 0 ==
    THEN 6 DROPN
'gA' RCL TAN 'o' RCL +
    END 100 ==
    THEN 5 DROPN
'gA' RCL 'p' RCL + TAN
    END 200 ==
    THEN 4 DROPN
'gA' RCL TAN 'o' RCL +
    END 300 ==
    THEN 3 DROPN
'gA' RCL 'p' RCL + TAN
    END 400 ==
    THEN DROP2 
'gA' RCL TAN 'o' RCL +
    END 400 >
    THEN DROP 
'gA'RCL 400 - '\GDa' STO
DUP
 IF '\GDa==0 OR \GDa==200'
      THEN 
'\GDa' RCL TAN 'o' RCL +
      ELSE 
'gA' RCL 'p' RCL + TAN
      END
    END 400 <
    THEN 'gA' RCL TAN
    END
  END '\GDa' STO 'gC'
RCL DUP DUP DUP DUP
DUP DUP DUP
  CASE 0 <
    THEN 7 DROPN
'gC' RCL 200 + TAN
    END 0 ==
    THEN 6 DROPN
'gC' RCL TAN 'o' RCL +
    END 100 ==
    THEN 5 DROPN
'gC' RCL 'p' RCL + TAN
    END 200 ==
    THEN 4 DROPN
'gC' RCL TAN 'o' RCL +
    END 300 ==
    THEN 3 DROPN
'gC' RCL 'p' RCL + TAN
    END 400 ==

    THEN DROP2 
'gC' RCL TAN 'o' RCL +
    END 400 >
    THEN DROP 
'gC' RCL 400 - '\GDc' STO
DUP
 IF '\GDc==0 OR \GDc==200'
      THEN 
'\GDc' RCL TAN 'o' RCL +
      ELSE 
'gC' RCL 'p' RCL + TAN
      END
    END 400 <
    THEN 'gC' RCL TAN
    END
  END '\GDc' STO 
'(\GDy - INV(\GDc)*\GDx) *
INV(-INV(\GDc)+INV(\GDa))'
\->NUM '\GDXm' STO 
'\GDXm * INV(\GDa)' \->NUM
'\GDYm' STO 
'Xa' RCL '\GDXm' RCL + 
'Xm' STO 
'Ya' RCL '\GDYm' RCL + 
'Ym' STO -16 CF 
'Xm' RCL 'Xm' \->TAG 
'Ym' RCL 'Xm' \->TAG 
{ \GDXm \GDYm \GDx \GDy \
GDa \GDc } PURGE 
{ AB AGT CD CGT ZERO } 
ORDER
\>>

5.2 Coordonnées XYZ 
Nouvelle demande, aucun code existant.

6 Annexe A : Procédures d’installation

6.1 Installation via connexion USB 

6.1.1 Connexion USB

Pour établir une connexion entre la HP50g et un PC (sous Windows), il faut, sur le PC : 

1. Installer le programme Conn4x  (HP48g,49g,50g series Calculator Connectivity Kit) disponible sur internet 
ou auprès d’HP.

2. Installer HP Graphing Calculators - USB Setup, disponible sur le site HP :
https://support.hp.com/us-en/drivers/selfservice/hp-50g-graphing-calculator/3235173

3. Connecter la HP50g via un câble USB – B
4. Mettre la calculatrice en mode serveur Xmodem (Xmodem Server) : pressez sur [Maj. Droite] 

[Flèche droite]), le message « Xmodem Server   Waiting for command » doit apparaître. 
Sortir de ce mode avec [CANCEL] (touche [ON]).

5. Cliquer sur le bouton Connect de Conn4X.
6. …

Après cette configuration et tests préliminaires, vous devez préparer les répertoires avant le transfert en soi.

6.1.2 Préparation de la HP50g

Avant d’effectuer les transferts, il faut créer la structure des répertoires et sous répertoires à la main :

HOME
+ GEOM
  + Droites
  + Cercles
  + Surface
  + XYZ

À partir de HOME :

 Accéder au gestionnaire de ficher (File Manager) : [FILES] ([Maj. Gauche] [APPS])
 Sélectionner le répertoire Home ([Up]/[Down] et OK)
 Créer un sous-répertoire GEOM :

o Accéder aux fonctions suivantes par [NXT] et ensuite NEW
o Laisser le champ Object vide, descendre sur Name et y entrer le nom du sous-répertoire 

[ALPHA] [ALPHA] GEOM [ALPHA] OK
o Sélectionner Directory (CHK) et OK.

 Revenir aux fonctions précédentes [PREV] ([Maj. Gauche] [NXT]), et TREE pour réafficher l’arbre des
répertoires, sélectionner GEOM, et y créer les sous-répertoires Droites, Cercles, Surface et XYZ ([NXT] 
NEW … (utilisez la [Maj. Gauche] pour obtenir les lettres minuscules)

6.1.3 Transfert des programmes

Une fois les répertoires créés, pour transférer les programmes, il suffit de les glisser/déposer depuis une fenêtre 
Windows Explorer vers le répertoire correspondant sous Connex4, répertoire par répertoire (un à la fois).

6.2 Sauvegarde des programmes dans la mémoire FLASH ou sur carte SD

Les programmes d’un répertoire et ses sous-répertoires peuvent être simplement copiés en mémoire FLASH ou 
sur une carte SD. Pour éviter de copier les variables (créées automatiquement lors de l’initialisation), il est 
préférable de réaliser les sauvegardes dès que les programmes on été transférés (via une connexion USB), copiés
(à partir d’une carte SD) ou tapés à la mains (sans erreurs !).

6.2.1 Mémoire FLASH

Pour la copie vers la mémoire FLASH (permanente, interne), vous devez :

 Accéder au gestionnaire de ficher (File Manager) : [FILES] ([Maj. Gauche] [APPS])
 Sélectionner le répertoire Home (avec les flèches) et OK
 Sélectionner ensuite le répertoire GEOM (avec les flèches) et COPY
 Sélectionner la destination 2:FLASH et OK
 [CANCEL] ([ON]) pour sortir et revenir en mode de calcul normal.

6.2.2 Carte SD

Prenez une carte de petite capacité, le maximum serait de 2 GB.

Selon la version de code de la HP50g (tapez [ALPHA] [ALPHA] VERSION [ENTER]), la carte SD doit être 
formatée en FAT32 (Unit size 4096B) pour la Revision #2.15 ou FAT16 si la version est antérieure. Cette 
opération peut se faire sur PC ou via le gestionnaire de ficher (File Manager) : [FILES] où la fonction FORMA est 
affichée dans le menu lorsqu’une carte SD est insérée (sous la touche [ . ], face vers le bas (contacts en haut)).

La procédure pour copier est identique à celle vers la mémoire FLASH, il faut, au point 4 :

 Sélectionner la destination 3:SD et OK

6.3 Ré-installation à partir d’une copie en mémoire FLASH ou sur carte SD
Il est également possible de sauvegarder uniquement un répertoire (et ses sous-répertoires), un programme ou 
une variable isolée sur la mémoire FLASH (mémoire permanente) ou sur une carte SD. C’est la méthode que j’ai 
choisie, et ai donc créé une copie de sauvegarde « prête à l’emploi » dans ces deux mémoires (FLASH & SD). 

NB : Une exécution de CFG peut s’avérer nécessaire après une copie depuis la carte SD ou la mémoire FLASH.

6.3.1 Carte SD

Avec la carte SD d’installation / de sauvegarde insérée (sous la touche [ . ], face vers le bas (contacts en haut)) :
 Accéder au gestionnaire de ficher (File Manager) : [FILES] ([Maj. Gauche] [APPS])
 Sélectionner la carte SD (avec les flèches) : 3:SD et [OK]
 Sélectionner le programme INST de la carte SD
 Exécuter directement ce programme, depuis la carte SD : [EVAL]
 Confirmer l’installation : [1] [ENTER] (toute autre entrée interrompt la procédure), le programme efface 

d’abord les éventuels répertoires et sous-répertoires de GEOM et copie de sauvegarde dans la mémoire 
FLASH avant de les (re)créer. Il initialise ensuite la calculatrice au moyen du programme CFG.

 Terminer par [CANCEL] et [APPS] pour activer le menu.

6.3.2 Mémoire FLASH

Procédure similaire à celle pour une ré-installation à partir d’une carte SD, la source change (point 2) ainsi que le 
programme à exécuter (point 3) :

 Accéder au gestionnaire de ficher (File Manager) : [FILES] ([Maj. Gauche] [APPS])
 Sélectionner la mémoire FLASH (avec les flèches) : 2:FLASH et [OK]
 Sélectionner le programme REST de la mémoire FLASH
 Exécuter directement ce programme, depuis la mémoire FLASH: [EVAL]
 Confirmer l’installation : [1] [ENTER] (toute autre entrée interrompt la procédure), le programme efface 

d’abord les éventuels répertoires GEOM /… et restaure les programmes à partir de la sauvegarde dans la 
mémoire FLASH. Il initialise ensuite la calculatrice au moyen du programme CFG.

 Terminer par [CANCEL] et [APPS] pour activer le menu. 

6.4  (Re-) Configuration de la HP 50g

Pour fonctionner correctement, les programmes GEOM supposent une configuration particulière de la calculatrice 
(voir section 2.2 : Configuration de la calculatrice HP 50g), automatiquement réalisée par le programme CFG.

A partir du menu de l’application GEOM [APPS] sélectionner l’option (répertoire) GEOM et ensuite le programme 
CFG (configuration) qui se trouve plus « à droite » (presser [NXT] jusqu’à ce qu’il apparaisse dans la ligne de 
menu). 

7 Annexe B : En cas de problème …

7.1 Calculatrice bloquée
Si la calculatrice est « bloquée », vous pouvez tenter :

1. Annulation de la séquence entrée (de la frappe) : : maintenir la touche [ON]enfoncée et presser sur [F2]
2. Redémarrage « à chaud », avec préservation des données (“Warm” restart) : maintenir la touche 

[ON]enfoncée et presser sur [F3]
3. Redémarrage « à froid », avec perte des données (“Cold” restart) : maintenir la touche[ON]enfoncée et 

presser ensuite sur [F1] et [F6] et relâcher dans l’ordre inverse. Vous pouvez tenter de récupérer la 
mémoire (YES ([F1])) ou pas (NO ([F6])).

4. Une réinitialisation (reset) en pressant le bouton RESET au dos de l’HP50g : enfoncez un trombone dans 
le trou « RESET ».

7.2 Calculatrice dé-configurée
Si, après une installation initiale, les programmes ne fonctionnent plus correctement, parce que la configuration de 
la calculatrice n’est plus correcte, certains programmes ont été corrompus (accidentellement modifiés, …) ou 
perdus, …  tentez les opérations suivantes, dans l’ordre (chacune étant plus «forte» que la précédente) :

1. Vérifier que la calculatrice est en mode utilisateur (indicateur USR dans la seconde ligne d’état), afin que la 
touche [APPS] active le menu du répertoire courant (raccourci pour [CUSTOM] : [Maj. Gauche] 
[MODE]). Pour l’activer, presser [USER] ([Maj. Gauche][ALPHA])

2. Exécuter le programme CFG du répertoire GEOM : reconfiguration et réinitialisation des variables.
3. Exécuter le programme REST (restaure) du répertoire GEOM : réinstalle les programmes à partir de la 

mémoire FLASH
4. Si le programme REST du répertoire GEOM n’existe plus ou ne fonctionne plus, exécuter la copie depuis la 

mémoire FLASH : procédure similaire à l’installation à partir de la carte SD : [FILES]  2:FLASH [OK] 
REST [EVAL]

5. Si la mémoire FLASH ne contient plus le programme REST et/ou la copie du répertoire GEOM, effectuer une
réinstallation complète à partir de la carte SD, via le programme INST disponible dans le répertoire GEOM 
et sur la carte elle-même (voir point 2.1)

6. En dernier recours, réinitialiser complètement la calculatrice (“Cold” restart). 
Attention : TOUTES LES DONNEES SERONT PERDUES !  

a. Maintenir la touche [ON]enfoncée, et presser ensuite simultanément sur [F1] et [F6]. 
b. Relâcher les touches dans l’ordre inverse
c. Appuyer sur [NO] ([F6]) pour ne pas récupérer la mémoire, puis [OK]
d. Réinstaller à partir de la mémoire FLASH ou de la carte SD.

